SPECIAL PRINT

INTERNATIONAL SYMPOSIUM ON ROCK MECHANICS RELATED TO DAM FOUNDATIONS

SLOPE STABILITY WITH PLANE, WEDGE AND POLYGONAL SLIDING SURFACES

KALMAN KOVÁRI and **P. FRITZ**
Federal Institute of Technology Zurich
ETH - Hönggerberg
CH - 8039 Zurich
Switzerland

September, 27 to 29, 1978
Rio de Janeiro, Brazil

Summary - Résumé :

Slope stability for planar, polygonal and wedge-shaped sliding surfaces: A formula for the design of anchors in rock slopes is presented. Since several parameters can be lumped together in two factors and the cohesion appears in an explicit form the formula can be used to carry out parametric studies with little expenditure of time. Due to an analogy between plane failure and the sliding of a wedge on two planes the same formula can be used to describe the three dimensional problem. An extension of the work to problems involving polygonal sliding surfaces is presented, which is based upon the hypothesis that due to kinematic considerations a discrete number of internal slip surfaces must exist in the rock mass. These internal slips may take place on preferred surfaces, which arise from the actual geological situation. In the majority of cases, however, new ruptures are created, that depend only partially on existing planes of weakness. In the numerical procedure the rock mass is divided up into discrete elements governed by the assumed internal slip surfaces. The basic formula mentioned above is then applied to each element separately. The internal forces acting on the interfaces between elements are defined by an additional failure condition. In an example taken from rock engineering practice it is shown how great the influence of the rock mass in resisting the development of such internal slip surfaces is with respect to the stability of the slope.

Stabilité de talus rocheux sur des surfaces de glissement polygonales et spaciales: Pour dimensionner l'ancrage d'un talus rocheux, on présente une formule qui permet d'effectuer des analyses paramétriques avec un travail minimum du fait que plusieurs paramètres sont résumés dans deux facteurs et que la cohésion est prise en compte explicitement. Grâce à une analogie entre le problème plan des talus et le glissement d'un coin de roche sur deux plans, le problème spacial se laisse aussi traiter avec la même formule fondamentale. Un élargissement du domaine d'application à des problèmes à surfaces de glissement polygonales se base sur le fait que pour des raisons cinématiques il existe des glissements internes ou cisaillements dans la masse rocheuse. Ces surfaces de cisaillement sont parfois données par la nature, mais la plupart du temps il se forme des cassures nouvelles qui ne suivent que partiellement des zônes faibles préexistantes. Les surfaces de cisaillement divisent la masse rocheuse en différentes parties où la formule fondamentale citée peut être appliquée séparément.

1. Introduction

Normally, for investigating numerically the stability of rock slopes, simplified deformation mechanisms are assumed. The computations are then carried out for various material properties and loading values. With the aid of the simple mathematical relationships presented here numerous behaviour hypotheses can be easily tested, so that a better understanding of the interplay of the forces in the rock structure is possible. Only on the basis of the knowledge thus gained and by considering influences not directly quantifiable can decisions be made regarding shape, drainage and safety measures in a rock slope. For a parameter analyses in rock engineering it is important that the extent of the mathematical formalism is kept to a minimum. The computational method given here meets this requirement in two respects. Firstly, a basic formula for the safety factor of rock slopes is presented, which is valid for both slip along a plane surface and of a three-dimensional rock wedge. This is possible thanks to the discovery of a formal analogy between the two problems. Secondly, the use of this formula is further simplified by means of charts or programming for a pocket calculator. The investigation of problems with polygonal slip surfaces gives a useful insight into the relationships holding for kinematically complex slides. Here, especially, the significance of potentially new failure surfaces, i.e. slip surfaces within the sliding rock mass, is evident.

2. Theoretical Foundations

The mathematical treatment of rock slides is based upon the hypothesis of limit equilibrium. The rock is idealized as a rigid body, and only sliding but no rotation or lifting-off of the potential sliding mass is considered.

2.1 Remarks on the defintion of safety factor

In civil engineering safety factor is usually understood as the relationship between the applied stress and a strength. In this sense the safety factor with respect to sliding of a slope is formulated as

$$F_s = \frac{\text{maximum shear resistance}}{\text{applied shear force}} \quad . \tag{1}$$

Another defintion that is frequently used is based upon a grouping of the forces acting on the sliding mass. This leads to the following definition of safety factor

$$F_s = \frac{\text{resisting forces}}{\text{driving forces}} \quad . \tag{2}$$

With the aid of a simple example it is shown that the two defintions can lead to completely different results (Kovári and Fritz, 1976). A body resting on an inclined plane is loaded by its self-weight W and an anchor force T. The corresponding components parallel and normal to the sliding surface, W_s, T_s and W_n, T_n respectively, are shown in Fig.1. If the maximum shear force at the moment of slip is designated by S_{max}, then according to definition (1)

$$F_s = \frac{S_{max}}{W_s - T_s} \quad ,$$

so that the component of the anchor force T_s reduces the effective applied force in the denominator. The safety factor defined by (2), however, becomes

$$\bar{F}_s = \frac{S_{max} + T_s}{W_s}$$

i.e. the anchor force contributes to increase the resisting force in the numerator.



Fig. 1: Potential sliding mass with the components of self-weight W_s, W_n and the anchor force T_s, T_n

It is evident that difficulties arise in applying the second definition. Whereas a loose anchor is considered as a passive resisting force in the numerator, if one is consistent, a prestressed anchor whould be considered as an active force reducing the resultant driving force in the denominator.

A comparison of the two definitions of safety factor for a given case is shown in Fig.2. The ratio T_s/W_s is plotted as abscissa, the ordinate representing the safety factor. Firstly, it is noticeable that the asymptotic value of F (Definition 1) at $T_s = W_s$ is infinite, since the resultant force parallel to the sliding surface disappears at this value, which means that failure can never occur. For even higher values of T_s the rock mass will slide upwards and F_s decreases. The second definition, however, is not capable of taking into account these two phenomena. For T_s=W_s the value of F_s is finite. Furthermore upwards movement of the rock mass is not accounted for. Only for downwards sliding in the range of values of F_s around 1 is there agreement between the two definitions. It is clear that the second definition



Fig. 2: Safety factor according to two different definitions

based on the consideration of driving and resisting forces should not be applied for the following reasons:

- upwards sliding is not accounted for,
- meaningless values are obtained for $T_s = W_s$.
- it infringes upon an elementary law of mechanics, according to which a group of forces acting on a rigid body is equivalent to its resultant. The determination of a resultant is not possible, because the external forces due to inadmissible grouping act partly as driving and partly as resisting.

2.2 Plane failure surface

A vertical section through a potentially unstable rock slope is shown in Fig.3. The weight of the rock mass is W and the contact area with the underlying rock on which it rests is A. The resultant R of all external forces acting (anchor force, water pressure, surcharge etc.) is directed at an angle β to the horizontal.



Fig. 3: Geometry of the slope and the forces acting on it

The geometry of the slope enters the calculation through the area A, the slope angle α and indirectly through the self-weight W. The reaction on the failure surface is composed of a normal force N and a shear force S. Resolving the forces into components parallel and normal to the failure surface leads to the following equilibrium equations

$$S + R \cos (\alpha + \beta) - G \sin \alpha = 0.$$

$$N - R \sin (\alpha + \beta) - G \cos \alpha = 0.$$

The definition of safety factor according to (1) is

$$F_s = S_{max} / S_{acting}$$

and Coulomb's failure condition for the contact surface is

$$S_{max} = N \tan \Phi + c A$$ (where c: cohesion, Φ: friction angle).

Thus the sought for basic formula (Kovari and Fritz, 1975), which represents the key to the simple treatment of slope stability problems, is obtained directly, namely

$$\boxed{R = k_1 \left(1 - \frac{cA}{W} k_2\right) W} \qquad (3)$$

The coefficients k_1 and k_2 are given by the expressions below

$$k_1 = \frac{F_s \sin\alpha - \cos\alpha \tan\Phi}{F_s \cos(\alpha+\beta) + \sin(\alpha+\beta) \tan\Phi}$$

$$k_2 = \frac{1}{F_s \sin\alpha - \cos\alpha \tan\Phi} .$$

The following points are worthy of note with regard to the above formula:
- if, besides the anchor force, no other external forces are acting, it may be used directly as a design formula for the anchor force.
- the cohesion c appears explicitly, which allows a very simple estimate of its influence to be made.
- it is valid, as will be shown afterwards, also for the three dimensional problem of sliding of a wedge on two plane surfaces.

The simple evaluation of this formula in practice is aided by the use either of a programmable pocket calculator or of design charts (Kovari and Fritz, 1976), whereby the factors k_1 and k_2 are a function of geometry, safety factor and friction angle only. An extension of this formula is described in appendix 4.1.

2.3 Sliding of a wedge on two planes

The three-dimensional problem is treated here of a wedge sliding on two planes with the contact areas A_1 and A_2. It is shown (Fig.4a) how this case is analogous to the simple one previously handled. We define a cartesian coordinate system (s,n,h). The s-axis lies in the direction of the line of intersection of the two planes, the h-axis is horizontal and the n-axis is in the vertical plane through the line of intersection. Fig.4b shows a section with a vertical plane through the intersection line, while Fig.4c shows a section normal to the intersection line.

The forces acting are divided into three groups:
- self-weight W.
- reaction (now two normal (N_1, N_2) and two shear forces (S_1, S_2) respectively).
- the resultant R of the external forces.

Analogue to the case of one sliding plane the basic relationships may be formulated for
- the three equations of equilibrium,
- the definition of safety factor (1),
- Coulomb's law of friction for shear resistance along the sliding surface.

To begin with, it will be assumed for the sake of simplicity, that the resultant R lies parallel to the vertical plane (s,n) through the intersection line, and further that the same friction angle applies to both planes. The combination of the five conditions with the help of elementary algebraic operations yields the following basic equation for wedge problems

$$\boxed{R = k_1^* \left(1 - \frac{c_1 A_1 + c_2 A_2}{W} k_2^*\right) W} \qquad (4)$$

where

$$k_1^* = \frac{F_s \sin\alpha_s - \cos\alpha_s \tan\Phi^*}{F_s \cos(\alpha_s+\beta) + \sin(\alpha_s+\beta)\tan\Phi^*}$$

$$k_2^* = \frac{1}{F_s \sin\alpha_s - \cos\alpha_s \tan\Phi^*}$$

$$\tan\Phi^* = \frac{\cos\omega_1 + \cos\omega_2}{\sin(\omega_1+\omega_2)} \tan\Phi = \lambda \tan\Phi .$$

One immediately recognizes the correspondence of this formula with that for a single plane and the conditions of the analogy:
- instead of the slope angle α of the single failure surface the slope angle α_s of the line of intersection of the two failure surfaces is employed,
- instead of the friction angle Φ the angle Φ^* is used,
- instead of the product cA the sum $c_1A_1 + c_2A_2$ must be considered.

The inclination α_s of the intersection line and the factor λ for determining the angle Φ^* may be read directly from tables for various geometries (Kovári and Fritz, 1976). In the same work all the necessary derivations are given in detail, especially for the angles ω_1 and ω_2.



Fig. 4: Isometric view and sections of a rock wedge



Fig. 5:

Rock slope in Canton Grisons, Switzerland

2.4 Example from rock engineering practice

With the aid of a practical example it will be shown how a parametric study can be carried out using the basic formula presented here. In the course of the reconstruction of a mountain road in Switzerland a section, approximately 100 m long, of a steep rock slope slid down (Fig.5). Since sliding of the remaining part of the slope was feared the use of rock anchors as a possible remedial measure was to be investigated. Characteristic for the whole rock mass was intensive folding and shearing of the interchanging beds of limestone and argillaceous shale (Fig.5). The most important consideration regarding potential instability was the presence of shear surfaces in the plane of the axis of folding, inclined at an angle 30°-40° normal to the road. The rock mass could break free from one of a number of joint systems running through it. Based on this data the problem was reduced to one of sliding on a single plane, whereby strips 1 m wide were considered. The endangered rock mass was divided in cross-section (see Fig.6) more or less arbitrarily into three parts, with the idea that the safety factor was best considered, for either just the lower part breaking off or the whole rock mass coming down.



Fig. 6: Typical cross-section through the potential sliding mass

In a first computational step the material properties of the sliding surface were back-calculated from the stability of the whole rock mass with the help of the formula (3). Substituting a safety factor $F_s = 1$ and various friction angles Φ the necessary cohesion c was calculated and plotted in Fig.7 in function of Φ. Each point on this curve represents a possible combination of values of c and Φ satisfying the condition of limit equilibrium.
For typical pairs of values of c and Φ the required anchor force was determined for various values of safety factor. For example, in Fig.8 the anchor force for three cases $W = W_1$, $W = W_1+W_2$ and $W = W_1+W_2+W_3$ is shown for a selected pair of values (c,Φ). If a smaller probability of occurrence is attributed to the whole rock mass (i.e. $W = W_1+W_2+W_3$), the safety factor $F_s = 1.1$ may be regarded as adequate. The corresponding anchor force $T \simeq 25$ t/m' means, however, that for sliding of the lower part



Fig. 7: Strength parameters for the limit equilibrium method (parameter back-calculation)



Fig. 8: Influence of the safety factor on the required anchor force

($W = W_1$) alone the safety factor is $F_s = 1.4$. The increase of the safety factor compared with that for sliding of the whole rock mass is in agreement with the greater probability that was assigned to this case.

With the simple estimate for the anchoring costs P in dependence upon the number of borings n, the anchor lengths l and the price P per unit of the anchors and P_B of the borings, i.e.

$$P = (P_A + n\ P_B)\ l$$

an optimum anchor inclination β was finally determined. The anchor costs in percent of the minimum value are plotted as ordinate against the angle β as abscissa in Fig.9. Depending upon the desired safety factor F_s there result, naturally, different anchor costs. However, the optimum inclination β for this example always lies between 10° and 30°.



Fig. 9: Determination of the most economic anchor inclination

3. Sliding on polygonal sliding surfaces

From experience it is known that sliding, corresponding to the nature of rock, usually takes place on polygonally-shaped surfaces. For such cases Janbu (1954) and Morgenstern and Price (1965) have suggested practical methods of computation, whereby the endangered earth or rock mass is divided up into vertical strips or slices. The computational procedure is based on certain assumptions regarding the distribution and slope of internal contact forces, as well as the hypothesis of limit equilibrium. The method advocated here, however, is based upon the physical requirement that sliding on a polygonal surface is only possible kinematically if a sufficient number of internal shear surfaces can develop. For the sake of simplification, in the following only continuous plane shear surfaces starting from the intersection lines of the polygon sliding surface are assumed. Thus, as shown in Fig.10, the slide of a mass on three planes must be accompanied by at least two internal shear surfaces. For n external sliding planes (n-1) such interfaces are required. The method described here rests upon the following basic assumptions:

a) The blocks comprising the rock mass are each considered to be rigid.

b) The directions of the internal shear surfaces are known.

c) On the internal and external sliding surfaces (at the condition of limit equilibrium) the Coulomb failure condition applies, and no tensile strength is permitted. The strength parameters may be allocated different values on each sliding surface.

d) For the safety factor - according to definition (1) - the same value is assumed for all internal and external sliding surfaces.



Fig. 10: Kinematics of a slope failure for a polygonal sliding surface

On the basis of these assumptions the safety factor and all external and internal reactions can be determined for a given geometry, loading and strength. The direction of the internal shear surfaces is chosen from case to case on the basis of a careful investigation of the structure of the potential sliding mass. However, for highly jointed rock the direction of the internal slip surfaces can be found by the condition of a minimum safety factor for the system. In an investigation of the stability of an earth dam Sultan and Seed (1967) used a similar criterium. It is readily seen that the resistance of a rock mass to splitting up into various parts plays an important role in stability calculations. The significance of interlocking effects in the joints and the strength of the rock are in this connection of great importance. Müller (1962, p.270) has already drawn attention to these aspects. With regard to the same safety factor being postulated for all slip surfaces the following remarks are offered. As will be shown in the next section, it would be formally possible to allocate a different safety factor to each slip surface. One limiting condition must be observed, i.e. that at the moment of slip the safety factor on all surfaces must be reduced to the value of unity. This refinement allowing for varying safety factors, however, seems to us, due to insufficient foundation and considering the many simplifications introduced to solve the problem, to be inappropriate. In any case, with considerations of this kind the contributions of the relative displacements along the interfaces, which are necessary to mobilize the shear resistance, must also be taken into account. One could quite easily imagine a situation, in which because of large deformations the external slip surface has reached a state of residual shear strength, whereas the internal slip surfaces (with smaller relative deformations) still exhibits the peak value of shear resistance. This consideration, which is related to the problem of progressive failure, exceeds the limits, however, of this present study. Indeed, the method of limit equilibrium, due to the assumption of rigid body behaviour, is not suitable to solve the problem of progressive failure. It is only possible to determine an admissible velocity field in the sense of the plasticity theory of Hill (1955).

3.1 Polygonal failure surface consisting of several planes

The general case of a potential rock slide on a n-section polygonal slip surface is shown in Fig.11. The geometry of the slope is fixed by the angles of inclination α_i and γ_i and the corresponding areas A_i and $\bar{A}_i$ of the respective slip planes. The forces acting are again divided into three groups:

- The weights W_i of the individual blocks,
- the external reactions N_i, S_i and the inner reactions $\bar{N}_i$, $\bar{S}_i$ (contact forces),
- the resultant R_i (of slope β_i) of the external forces (anchor force, water pressure in the external slip surfaces etc.). Water pressure, that may act in the joints normal to the internal slip surfaces, are taken care of by the forces $\bar{W}_i$.

If it is now assumed that there is a different safety factor in each slip plane, four unknowns are obtained for each plane, the safety factor, two reaction forces and one strength value (S_{max}). Thus for n external and (n-1) internal slip surfaces there are altogether (8n-4) unknowns to be determined. For this purpose, for each of the n blocks, two equilibrium conditions must be satisfied

$$\Sigma X_i = 0, \quad \Sigma Y_i = 0$$

as well as the [n + (n-1)] Coulomb conditions on the sliding planes and the corresponding expressions (1) for safety factor in the form

$$S_{i_{max}} = N_i \tan\Phi + c_i A_i \ (N_i \geq 0), \quad \bar{S}_{i_{max}} = \bar{N}_i \tan\bar{\Phi} + c_i A_i \ (\bar{N}_i \geq 0)$$

$$F_{s_i} = \frac{S_{i_{max}}}{S_i}, \quad \bar{F}_{s_i} = \frac{\bar{S}_{i_{max}}}{\bar{S}_i}$$

In the above it was tacitly made use of the principle of action and reaction for the component forces $\bar{N}$,$\bar{S}$ on the interface. Thus for the system as a whole with n blocks there are (6n-2) equations and (8n-4) unknowns, i.e. the problem is statically indeterminate to the (2n-2)th degree. This indeterminacy is a consequence of the working hypothesis of the method of limit equilibrium, as with the assumption of rigid behaviour the displacement and stress fields are unknown. The deficient equations, therefore, cannot be found using mechanical or physical laws. A possible hypothesis is to make the safety factors in the various slip planes dependent upon one another. Since the safety factor in the case of slip must be everywhere unity and here a simplified approach is sought, we assume that the safety factor is equal in all slip planes. The deficient equations are thus

$$F_{s_i} = F_{s_1} \qquad (2 \leq i \leq n+n-1).$$

In solving the system of equations it should be noted, that most of the unknowns can be eliminated with the help of the basic formulas (3) and (7) respectively, leaving just n values. To solve the remaining equations, on account of their nonlinear character, an iterative method is used. The equations are best solved, therefore, by means of the computer program listed in appendix 4.2. For the special case of sliding on a two-degree polygonal surface it is shown in the next section that this problem can also be solved by hand with not too much effort. For the sake of completeness, a semigraphical procedure for the analysis of an n-degree polygonal sliding surface is also presented. Generally, this procedure would only find application if an electronic computer were not available.
Remark: If, with the exclusion of tensile strength of rock, negative contact (interaction) forces occur this points to a separation of the individual blocks. From the point of view of the stability analysis, the slide of the whole rock mass is no longer of interest, but only of a certain group of blocks.



Fig. 11: Element i of a rock mass with a polygonal sliding surface
a) geometrical quantities
b) internal and external forces

3.2 Polygonal sliding surface composed of two planes

The rock mass shown in Fig.12a rests on two potential sliding planes of angle of inclination α_1 and α_2 . Sliding is only possible if an internal slip surface with a certain inclination γ can develop, so that the mass is divided into two blocks of weight W_1 and W_2 respectively. Besides possible external forces R_1 and R_2 and a water pressure $\bar{U}$ acting in the interface there are the reactions N_1, S_1 and N_2 respectively on the external slip planes and an interaction force I on the internal slip surface. The components $\bar{N}$, $\bar{S}$ of I must fulfil the failure condition of Coulomb. With the parameters for the internal slip surface - cohesion $\bar{c}$, friction angle $\bar{\Phi}$ and contact area $\bar{A}$ - the Coulomb condition is



Fig. 12: Rock mass with forces acting on it for a sliding surface consisting of two planes

$$\bar{S}_{max} = (\bar{N} - \bar{U}) \tan\bar{\Phi} + \bar{c}\bar{A} .$$

Employing the definition (1) for safety factor, viz.

$$F_s = \frac{\bar{S}_{max}}{\bar{S}} = \frac{\bar{N} - \bar{U}}{\bar{S}} \tan\bar{\Phi} + \frac{\bar{c}\bar{A}}{\bar{S}}$$

one obtains the characteristic angle δ (Fig.12b) between the interaction force I and the normal to the corresponding sliding surface:

$$\tan\delta = \frac{\bar{S}}{\bar{N}} = \frac{1}{F_s} \left(\tan\bar{\Phi} + \frac{\bar{c}^*\bar{A}}{\bar{N}}\right) , \tag{5a}$$

with the relationships

$$\bar{N} = I \cos\delta \text{ and } \bar{c}^* = \bar{c} - \frac{\bar{U}}{\bar{A}} \tan\Phi .$$

As a known special case one obtains, for cohesionless material, without water pressure in the interface and a safety factor $F_S = 1$, the value of δ :

$$\delta' = \bar{\Phi} . \tag{5b}$$

The angle of inclination $\bar{\beta}$ of the interaction force follows from considering Fig.12b for the lower block 1, i.e.

$$\bar{\beta}_1 = \frac{3\pi}{2} - \gamma - \delta \tag{6a}$$

and for the upper block 2

$$\bar{\beta}_2 = \beta_1 \pm \pi . \tag{6b}$$

The safety factor for the whole rock mass is found from the condition that this value must be equal for both elements. In practice the necessary force I is found separately for each block with the aid of the basic formula (7) as a function of the safety factor. For the lower element 1 the following expression holds, for example:

$$I = k_1\left(1 - \frac{c^*A}{W_1+R_{1_w}} k_2\right) (W_1+R_{1_w}) - R_{1_i} .$$

The angle of inclination $\bar{\beta}$ of I is chosen, as a first approximation, with the help of eqns. (5b) and (6). The resultant force R_1 or its components R_{1_w} and R_{1_i} in the directions of W and I respectively represent external forces acting on the block such as surcharge, anchor force, water pressure in the external slip surface (water pressure in the internal slip surface is accounted for by a reduced cohesion $\bar{c}^*$). For the angle $\bar{\beta}$ and the assumed value of safety factor F_s the coefficients k_1 and k_2 and thus the interaction force I are now determined. Fig.13a shows the interaction force as a function of safety factor plotted separately for both blocks. The intersection of the two curves gives the required solution, because at this point both the interaction force and the safety factor are equal for the adjoining blocks. Since, however, the inclination $\bar{\beta}$ of the interaction force was only an initial trial value, it must be redetermined by substituting for F_{s_g} and I in (5a) and then the calculation must be repeated iteratively with the new value of δ. Usually the method converges very rapidly. A case of interest is shown in Fig.13b, in which the equilibrium of the two blocks, i.e. the point of intersection of the two curves leads to a negative interaction force. This corresponds to the situation, where the safety factor of the lower element 1 is less than that of the upper element. Due to the requirement that the interforce cannot transmit tensile forces it follows that the lower element alone is decisive for stability considerations with I = 0 and not elements acting together. For example, if F_{s_g} = 1 sliding only of the lower element is to be expected, while the upper element may remain stable.



Fig. 13: Effective safety factor F_{s_g} of a body on two planes of different inclinations

3.3 Semi-graphical method of solution for a polygonal slip surface composed of several planes

If an analysis has to be carried out for a multiple-plane slip surface and use is not made of the computer program given in the appendix, then it is also possible to obtain a solution by semi-graphical means. Essentially, this method is based upon a successive determination of the interaction forces I_i in the various blocks making use of the boundary condition $I_n = 0$. The following description should suffice to explain the detailed steps of the method: With the help of the basic formulae (3) or (7) the interaction force is found first for element 1 for various values of the safety factor F_s. Since the inclination β_1 of the interaction force in eqns (5) and (6a) is dependent upon I_1 this calculation is of an iterative nature. Using the graphical relationship for $I_1 = f\ (F_s)$ in Fig.14a one can find the interaction forces $I_1F_s^1$, $I_1F_s^2$, $I_1F_s^3$ corresponding to the three values of safety factor F_s^1, F_s^2, F_s^3 respectively. The second block is now subjected to applied forces $- I_1F_s^i$ in a successive manner and for the corresponding values of the safety factor, F_s^i, the necessary interaction forces $I_2F_s^i$ are found (Fig.14b). Due to the implicit representation of the inclination angle β_2 this calculation must also be carried out interatively. Afterwards, the third element is loaded with the forces $- I_2F_s^i$, and so on. From the condition that for the n-th element $I_n = 0$, the sought-for factor of safety F_{sg} is obtained by means of interpolation in Fig.14c.



Fig. 14: Determination of the safety factor for a polygonal sliding surface



Fig. 15: Highway section with failure mechanisms (A) and (B)

3.4 Example of a stability check of a rock mass resting on an polygonal slip surface

As part of a national highway project in Switzerland it was proposed to use prestressed anchors to stabilizie a section of highway cutting through rock. The rock mass, a chalky marl, was characterized in this area by pronounced bedding, dipping towards the road at a shallow angle. Due to the structure of the rock two different sliding mechanisms with polygonal slip planes were assumed (Fig.15). In case (A) the extent of the rock mass is such that the anchor forces are fully effective, whereas in case B the sliding surfaces were so deep that they were beyond the anchoring zones. For the upper sliding surface (case (A)) the influence of the inclination γ_2 of the interface on the safety factor was investigated. The safety factor corresponding to various values of γ_2 was evaluated (Fig.16), plotting γ_2 as ordinate and safety factor as abscissa. The minimum safety factor leads to the critical value of γ_2 for the safety factor F_{sg} of the rock slope. The strength parameters for the internal slip surface were assumed to be $\bar{c}$ = 1.0 kp/cm², and Φ = 25°. The second problem that had to be investigated in connection with this project was the stability with respect to a deep-seated failure surface (B). In a similar way to the previous case, the first step was to determine the critical inclinations of the internal slip surfaces. It was found that for the same strength parameters the same value of γ_2 was obtained for the right interface (see Fig.15) as for case (A). The left slip surface with inclination γ_1 was determined (practically independently of strength properties) due to the constraint that it passes through the foot of the retaining wall. The influence of strength in the shear surface on the stability of the potential sliding mass is evident from Fig.17. For constant material properties in the external slip surfaces an increase of the cohesion of 2.0 kp/cm² in the internal slip surfaces effects an increase of safety factor $\Delta F_s \simeq 1.0$. The influence of the friction angle $\bar{\Phi}$ by contrast is much smaller. For the purpose of comparing these results with other methods of analysis, the safety factor based on Janbu's method was also computed for case (B). If, in Janbu's method, the interface forces are neglected, a safety factor ν = 2.6 results. This value holds per definition independent of the interface parameters $\bar{c}$ and $\bar{\Phi}$. More "exact" computations using Janbu's method considering the interaction forces do not lead, for this example, to reasonable re-



Fig. 16: Influence of the inclination of the internal shear surface on the safety factor (case Ⓐ)



Fig. 17: Influence of the strength of the internal shear surfaces on the safety factor (case Ⓑ)

sults, as a variation of the assumed slope of the pressure line or of the interaction forces by just a few degrees already has the effect of doubling the factor of safety.

4. Appendix

4.1 Extension of the basic formula

Often the resultant R of all the external forces is composed of a force K whose magnitude and direction is known (e.g. water pressure) and of a force T whose direction β is known, but nor its magnitude (e.g. anchor force, interaction force). In order to use the basic formulae (3) and (4) here also directly for design purposes (i.e. to find T), the force K is resolved into its components K_t in the direction of T and K_w in the direction of gravity (Fig.18).



Fig. 18: Resolution of a force K into the componentes K_t and K_w

The resultant R is thus replaced in the basic formula by the value $R = T + K_t$. The effective weight increases to $W' = W + K_w$. Thus, in this instance, the modified form of equation (3) is

$$T = k_1 \left(1 - \frac{cA}{W+K_w} k_2\right) (W + K_w) - K_t . \qquad (7)$$

The coefficients k_1 and k_2 are the same as for eqn. (3).

4.2 Computer program for computation of polygonal sliding surfaces

The following simple program-subroutine, written in FORTRAN language, may be used to calculate the safety factor of a rock mass on a polygonal sliding surface. Geometry, material characteristics, and external forces are given as input data and the safety factor is calculated in function thereof, as well as the magnitude and direction of interaction forces. The input and output values are transferred by formal parameters. Their significance is explained at the head of the subroutine listing. The method of solution is based upon an extended form of eqn. (7). With the nomenclature of Fig.11 one obtains the interaction force for the i-th element, viz.

$$I_i = k_1\left(1- \frac{c_i A_i}{W_i+R_{i_w}+I_{i-1_w}} k_2\right) (W_i+R_{i_w}+I_{i-1_w}) - R_{i_I} - I_{i-1_I} \qquad (1 \le i \le n)$$

whereby I_n must equal zero (R_{i_I} and R_{i_w} etc. designate the components of R_i in the directions of I_i and W_i respectively). The general form of this equation leads to a non-linear system of equations of the n-th degree, which, with the help of an iterative process, is solved in linear form. Since the number of equations n to be solved is normally quite small, the computer costs are also very small.

```
      SUBROUTINE     POLY(N,ALPHA,CFG,PHIG,G,AR,BETAR,GAMA,CFS,PHIS,W,
     1                    SNUE,AK,BETAK,MODE                          )

CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C                                                                      C
C     SLOPE   STABILITY   OF   A   ROCK   MASS   ON   A                C
C     P O L Y G O N A L   S L I D I N G   S U R F A C E                C
C                                                                      C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C                                                                      C
C     INPUT       DATA    (ALL ANGLES IN RADIAN)                       C
C     ---------------                                                  C
C     N           NUMBER OF EXTERNAL SLIDING SURFACES                  C
C     ALPHA(N)    DIP ANGLE OF EXTERNAL SLIDING SURFACES               C
C     CFG(N)      COHESION IN EXTERNAL SLIDING SURFACES MULTIPLIED     C
C                           BY AREA OF SLIDING SURFACES                C
C     PHIG(N)     ANGLE OF FRICTION IN EXTERNAL SLIDING SURFACES       C
C     G(N)        WEIGHT OF INDIVIDUAL BLOCKS                          C
C     AR(N)       EXTERNAL FORCES ACTING ON SLIDING BLOCKS             C
C     BETAR(N)    ANGLE OF INCLINATION OF FORCES AR(N)                 C
C     GAMA(N-1)   DIP ANGLE OF INTERNAL SHEAR SURFACES                 C
C     CFS(N-1)    COHESION IN INTERNAL SHEAR SURFACES MULTIPLIED       C
C                           BY AREA OF SHEAR SURFACES                  C
C     PHIS(N-1)   ANGLE OF FRICTION IN INTERNAL SHEAR SURFACES         C
C     W(N-1)      WATER PRESSURE IN INTERNAL SHEAR SURFACES            C
C                           (UNITS OF FORCE)                           C
C                                                                      C
C     OUTPUT       DATA                                                C
C     ----------------                                                 C
C     SNUE        SAFETY FACTOR                                        C
C     AK(N)       INTERACTION FORCES IN INTERNAL SHEAR SURFACES        C
C     BETAK(N)    ANGLES OF INCLINATION OF FORCES AK(N) IN RADIANS     C
C     MODE =  0   COMPUTATION  OK                                      C
C          >  0   INTERACTION FORCES < 0                               C
C          = -1   ERROR: NO CONVERGENCE                                C
C                 CORRECTIVE MEASURE: DECREASE SFAKT           OR/AND  C
C                                     INCREASE TOLDELT,SPEDNUE,SPEDELT C
C          = -2   ERROR: DIVISION BY ZERO                              C
C          = -3   ERROR: EQUATION SYSTEM NOT SOLVABLE                  C
C                                                                      C
C     R E M A R K S                                                    C
C     -------------                                                    C
C     COMMON /MATRIX/ Z(N*N)    DIMENSIONING OF Z AND OF ALL           C
C                               ACTUAL PARAMETERS HAS TO BE CARRIED    C
C                               OUT IN THE CALLING PROGRAM             C
C                                                                      C
C                               P.FRITZ + K.KOVARI    ETH ZUERICH      C
C                                                                      C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC

      DIMENSION ALPHA(1), CFG(1), PHIG(1), G(1), AR(1), BETAR(1),
     1          GAMA(1),  CFS(1), PHIS(1), W(1), AK(1), BETAK(1)
      COMMON /MATRIX/ Z(1)
      DATA SPEDNUE,SPEDELT/0.5,0.75/,TOLDELT/0.05/,TOLNUE/0.5/,
     1     TOL/0.0001/,ITERMAX/100/,SFAKT,SMAX/1.5,50.0/
```

```
      IF(N.EQ.1) GOTO 35
      ATAN1 = ATAN(1.0)/45.0
      PI15 = 270.*ATAN1
      N1 = N - 1
      BETAK(N) = 0.0
      DO 2 I=1,N
    2 PHIG(I) = TAN(PHIG(I))
      DO 3 I=1,N1
      AK(I) = 0.
      PHIS(I) = TAN(PHIS(I))
    3 CFS(I)  = CFS(I) - W(I)*PHIS(I)
      SNUEO = 1./SFAKT
    5 ITER = 0
      SNUEO = SFAKT*SNUEO
      SNUE  = SNUEO
      IF(SNUEO.GT.SMAX) GOTO 40
C
C     COMPUTATION OF THE DIRECTION OF CONTACT FORCES
C
   10 KONV = -1
   11 ITER = ITER + 1
      IF(ITER.GT.ITERMAX) GOTO 41
      IF(SNUE.EQ.0.0) GOTO 42
      DO 19 I=1,N1
      VORZ = 1.
      IF(ALPHA(I+1).LT.ALPHA(I)) VORZ = -1.
      IF(ITER.EQ.1) DELTA = ATAN(PHIS(I)/SNUE)
      IF(ITER.GT.1) DELTA = PI15 - GAMA(I) - BETAK(I)
      IF(ITER.GT.1.AND.CFS(I).NE.0.0) GOTO 13
      TGDELTA = PHIS(I)/SNUE
      GOTO 15
   13 COSDELT = COS(DELTA)
      IF(COSDELT.EQ.0.C) GOTO 42
      TGDELTA = (PHIS(I)+CFS(I)/AK(I)/COSDELT)/SNUE
   15 DDELTA = SPEDELT*(ATAN(TGDELTA)-DELTA)
      DDABS = ABS(DDELTA)
      DMIN = AMIN1(DDABS,TOLDELT*DELTA)
      DELTA = DELTA + SIGN(DMIN,DDELTA)
      BETAK(I) = PI15 - GAMA(I) - DELTA*VORZ
   19 CONTINUE
C
C     SET UP LINEAR EQUATION SYSTEM WITH UNKNOWN
C     CONTACT FORCES AK AND CHANGE OF SAFETY FACTOR DSNUE
C
      DO 29 J=1,N
      NRZ = (J-1)*N
      IE = J - 2
      IF(J.LE.2) GOTO 22
      DO 21 I=1,IE
   21 Z(NRZ+I) = 0.
   22 F1Z = SNUE*SIN(ALPHA(J)) - COS(ALPHA(J))*PHIG(J)
      F1N = SNUE*COS(ALPHA(J)+BETAK(J)) + SIN(ALPHA(J)+BETAK(J))*PHIG(J)
      IF(F1Z.EQ.0.0.OR.F1N.EQ.0.0) GOTO 42
      F1 = F1Z/F1N
      F2 = 1.0/F1Z
      HILF = SIN(BETAR(J))-COS(BETAR(J))*TAN(BETAK(J))
      ARG = HILF*AR(J)
      IF(J.EQ.1) GOTO 24
      I = IE + 1
      COSBKJ = COS(BETAK(J))
      IF(COSBKJ.EQ.0.0) GOTO 42
```

```
      ARIG = ARG
      IF(J.GT.1) ARIG = ARIG + HILF*AK(J-1)
      HILF = G(J) + ARIG
      IF(HILF.EQ.0.0) GOTO 42
      F1K = F1*(1.-CFG(J)/HILF*F2)
      Z(NRZ+I) = F1K*(SIN(BETAK(J-1))-COS(BETAK(J-1))*TAN(BETAK(J)))
     1           - COS(BETAK(J-1))/COSBKJ
   24 CONTINUE
      IF(J.EQ.N) GOTO 28
      Z(NRZ+J) = 1.
      IF(J.EQ.N1) GOTO 28
      IA = J + 1
      DO 26 I=IA,N1
   26 Z(NRZ+I) = 0.
   28 DF1 = (SIN(ALPHA(J))*F1N-COS(ALPHA(J)+BETAK(J))*F1Z)/F1N/F1N
      DF2 = -SIN(ALPHA(J))/F1Z/F1Z
      HILF = G(J) + ARG - F2*CFG(J)
      Z(NRZ+N) = -DF1*HILF + DF2*F1*CFG(J)
      ARK = COS(BETAR(J))/COS(BETAK(J))*AR(J)
   29 AK(J) = F1*HILF - ARK
C
C     SOLVE EQUATION SYSTEM AND CHECK CONVERGENCY
C
      CALL GAUSS(N,Z,AK,MODE)
      IF(MODE.NE.0) GOTO 43
      DO 33 I=1,N1
      IF(AK(I).LT.0.0) GOTO 5
   33 CONTINUE
      DSNUE = AK(N)
      DSN1 = ABS(SPEDNUE*DSNUE)
      DSN2 = ABS(TOLNUE*SNUE)
      DSN  = SIGN(AMIN1(DSN1,DSN2),DSNUE)
      SNUE = SNUE + DSN
      IF(ITER.EQ.1.OR.ABS(DSNUE/SNUE).GT.TOL) GOTO 10
      I = 0
      IF(KONV.EQ.0) GOTO 40
      KONV = KONV + 1
      GOTO 11
   35 SNUE = (AR(1)*SIN(ALPHA(1)+BETAR(1))+G(1)*COS(ALPHA(1)))
     1       *TAN(PHIG(1)) + CFG(1)
      SNUE = SNUE/(G(1)*SIN(ALPHA(1))-AR(1)*COS(ALPHA(1)+BETAR(1)))
      I = 0
C
C     TERMINATION
C
   40 MODE =  I
      GOTO 46
   41 MODE = -1
      GOTO 46
   42 MODE = -2
      GOTO 46
   43 MODE = -3
   46 AK(N) = 0.0
      IF(N.EQ.1) GOTO 49
      DO 47 I=1,N
   47 PHIG(I) = ATAN(PHIG(I))
      DO 48 I=1,N1
      CFS(I)  = CFS(I) + W(I)*PHIS(I)
   48 PHIS(I) = ATAN(PHIS(I))
   49 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE GAUSS(N,Z,B,MODE)

C
C     SIMPLE GAUSSIAN ALGORITHM (PIVOT ALWAYS IN DIAGONAL)
C
C     N              NUMBER OF EQUATIONS
C     Z(N*N)         COEFFICIENT MATRIX IN ROWS (IS OVERWRITTEN)
C     B(N)           VECTOR OF CONSTANTS (IS OVERWRTITTEN BY
C                    VECTOR OF RESULTS)
C     MODE           CONTROL PARAMETER SET BY GAUSS
C          = 0       SOLUTION OK
C          = 1       PIVOT = 0
C
      DIMENSION Z(1),B(1)

      N1 = N - 1
      DO 19 K=1,N1
      NRZO = (K-1)*N
      NPIV = NRZO + K
      IA = K + 1
      DO 19 J=IA,N
      NRZ = (J-1)*N
      NK = NRZ + K
      IF(Z(NPIV).EQ.0.0) GOTO 41
      S = Z(NK)/Z(NPIV)
      DO 18 I=IA,N
      NK = NRZ + I
      NZ = NRZO + I
   18 Z(NK) = Z(NK) - S*Z(NZ)
   19 B(J)  = B(J)  - S*B(K)

      DO 29 II=1,N
      I = N - II + 1
      NRZ = (I-1)*N
      S = B(I)
      IF(I.EQ.N) GOTO 25
      IA = I + 1
      DO 22 IZ=IA,N
      NZ = NRZ + IZ
   22 S = S - Z(NZ)*B(IZ)
   25 NZ = NRZ + I
   29 B(I) = S/Z(NZ)
      MODE = 0
      GOTO 46
   41 MODE = 1
   46 CONTINUE
      RETURN
      END
```

References

Hill, R., 1955, The mathematical theory of plasticity. Oxford University Press.

Janbu, N., 1954, Application of Composite Slip Surfaces for Stability Analysis. Proc. Europ. Conf. Stability of Earth Slopes, V.3.

Kovári, K., Fritz, P., 1975, Stability Analysis of Rock Slopes for Plane and Wedge Failure with the Aid of a programmable Pocket Calculator. 16th Symp. Rock. Mech., Minneapolis, USA.

Kovári, K., Fritz, P., 1976, Stabilitätsberechnung ebener und räumlicher Felsböschungen, Rock Mechanics 8. Int. Journal of ISRM, Springer.

Morgenstern, N.R., Price, V.E., 1965, The Analysis of the Stability of General Slip Surfaces. Geotechnique, V.15.

Müller, L., 1963, Der Felsbau. Verlag F. Enke, Stuttgart.

Sultan, H.A., Seed, B.H., 1967, Stability of sloping core earth dams. Journal Soil Mech. Found. Div., Proc. ASCE.